
CHART PATTERNS

Identify And Trade Formations

Detecting Breakouts
From Flags & Pennants

In

Here’s a system to identify and trade flag
and pennant formations.

by Markos Katsanos

my previous article, I wrote
about the measuring impli-
cations, duration, and iden-
tifying characteristics of
flags and their variations,
and promised to present a

system to identify and trade these short-
term and highly profitable formations. I
constructed a viable system by distilling
the most useful statistics together with tech-
nical observations of 100 flag and pennant
samples. I optimized the test parameters in
further out-of-sample testing of a list of
250 stocks.

In this article, I have presented the for-
mula and methods used in this system,
along with the test results on four stocks. I
found the most profitable test returned an
adequate 187% profit being only 8% of the
time in the market, versus a tragic 91% loss
for the buy and hold investor.

A frequent trader, however, will need to
wait for the next entry signal, since the
system test produced only one trade per
year on average. Nevertheless, this need
not be a major disadvantage, as a daily scan
of a large database of stocks is bound to
produce a few trades. I ran a MetaStock
exploration on a database of 1,250 stocks
and produced 13 hits. Results were highly
profitable, producing an annualized 735%
profit.

TEST DESIGN
The formula language of off-the-shelf tech-
nical analysis programs like MetaStock
might be easy to use for the usual tasks but
presents major limitations in recognizing
patterns or formations. On the other hand,
a lower-level programming language can
be more effective but will require exten-
sive and tedious programming.

M
AR

K 
 M

O
LN

AR



  Copyright (c) Technical Analysis Inc.

Stocks & Commodities V. 23:5 (48-57): Detecting Breakouts From Flags & Pennants by Markos Katsanos

Nevertheless, I did
my best to design a sys-
tem that identified an
adequate number of
flags and pennants, us-
ing only the formulas
available in MetaStock
(see sidebar 1,
“MetaStock code: sys-
tem test, flags & pen-
nants”). I did so with
only nine lines of code,
as opposed to more than
a hundred that would
have been required had
I used a lower-level pro-

Sell Order

D1:=Simulation.CurrentPositionAge; D2:=LASTVALUE(D1);
 zz:=Zig(Ref(C,-D2),17,%);
X:=BarsSince(ZZ<Ref(ZZ,-1)AND Ref(ZZ,-1)>Ref(ZZ,-2));
X1:=LastValue(X)+1; {flag duration}
c1:=LLV(C,(D2+40)); {pole base}
c2:=ref(hhv(c,22),-D2); {pole top}

BREAK1:= Simulation.CurrentPositionPerformance*100;
POLE:=(C2-C1)/C1*100; {pole height %}

{EXIT CONDITIONS}
BREAK1 >1.94* Power(pole ,.724) { Exit condition 1- profit target }
OR C<REF(LLV(C,X1),-D2) { Exit condition 2 stop loss on breaking
of lower flag trendline}
OR (D2>14 AND BREAK1<.25*POLE) { Exit condition 3-inactivity }
OR (D2>3 AND C<.9*HHV(C,4)) { Exit condition 4-trailing stop loss}
OR D2>24 { Exit condition 5-time exit}

FLAG & PENNANT STATISTICS
Formation Formation Volume Breakout Flagpole Formation

Breakout Flagpole Slope Slope% Decline% Duration Duration Duration
% Height % Total % Per Day Per Day Days Days Days

Average 45% 67% -3.4% -0.4% -9.5% 11 13 10
Median 33% 54% -2.7% -0.2% -5.1% 9 10 9
95% percentile 123% 160% 5.0% 0.6% -47% 24 29 21
90% percentile 87% 125% 3.0% 0.2% -25% 22 26 16
10% percentile 15% 24% -9.7% -1.2% 0.7% 3 3 5
5% percentile 12% 21% -13.1% -1.8% 2.0% 2 2 4

   To create this test, click on “Enhanced system tester,” click on
“New system,” and type the buy and sell order code shown above.
To run the test, click on “New simulation,” add securities, and select
the stocks you want to test. Then click on “Next,” type in an initial
equity of $10,000, select default size transaction cost 10000, select
“only long trades,” and check “Close all positions on the last bar.”
Click on “More…”, fill in an interest rate of 3%, and fill in 10 points
per transaction for the commissions. Click on “trade execution,”
uncheck “realistic market prices,” and select “Buy price at open”
and “Sell price at close.” Fill in one day for the delay.—MK

gramming language instead. I do not claim, however, that the
system can identify all possible flag patterns that could be
detected visually or otherwise.

I used mainly the zigzag and linear regression slope func-
tion to identify the patterns. The zigzag function identified
the flagpole top, and the linear regression slope ensured that
a steep and quick move was followed by a sideways or
slightly down flag formation. The following test conditions
were used to identify the pattern.

BUY CONDITIONS
Buy condition 1: Flag duration

The flag or pennant duration was limited to 21 trading days.
This was derived from the statistical results presented in my
previous article and reproduced here. According to Figure 1,
the maximum duration of 21 days included 95% of the flags
in the sample.

The duration was measured from the latest reversal iden-
tified by the zigzag function.

Buy condition 2: Flagpole
A 24% rise for the pole would include, according to the
statistics in Figure 1, 90% of the flags in the sample. How-
ever, after testing I found that this was not steep enough, as
some nonflag formations were identified incorrectly. So I

FIGURE 1: STATISTICS FOR BULLISH-ONLY FORMATIONS. The percent breakout is measured from the point at which the upper
trendline is broken in an uptrend. The duration is in trading days and does not include weekends and holidays. The formation
characteristics are divided by percentiles according to values below which certain percentages of cases fall. So for example, the 5%
percentile of the formation duration is four days, which means that 5% of the formations will last for four days or less.

METASTOCK CODE: SYSTEM TEST, FLAGS & PENNANTS

Enter Long

ZZ:=Zig(C,17,%);
X:=BarsSince(ZZ<Ref(ZZ,-1)AND Ref(ZZ,-1)>Ref(ZZ,-2));
X1:=LASTVALUE(X)+1; {flag duration}
X2:=X1+1;
SD:=Stdev(C,X2);

PERIOD:=22;
COEF:=.1;
INTRA:=Log(H)-Log(L);
VINTRA:=Stdev(INTRA,PERIOD);
INTER:=Log(Typical())-Log(Ref(Typical(),-1));
VINTER:=Stdev(INTER,PERIOD);
CUTOFF:=COEF*(VINTER+VINTRA)*C;
MF:=C-(H+L)/2+Typical()-Ref(Typical(),-1);
FVE:=Sum(If(MF>CUTOFF, +V, If(MF <-CUTOFF, -V,0)),PERIOD)/
Mov(V,PERIOD,S)/PERIOD*100;

X1<21 and X1>2 AND {Condition 1}
Ref(LinRegSlope(C,13)/Ref(C,-13)*100,-X1)>2.2{Condition 2}
AND Ref(LinRegSlope(C,X2)/Ref(C,-X2),-1)*100<.2
AND LinRegSlope(C,X1)/Ref(C,-X1)>-1.2 {Condition 3}
AND Ref(LinRegSlope(V,X2)/Ref(V,-X2),-1)*100<-2 {Condition 4}
AND Ref(LinRegSlope(SD,X1),-1)<0 {Condition 5}
AND Stoch(20,3)>55 AND ADX(10)>30{Condition 6}
AND FVE>10 AND Fml(“VFI”)>-3 {Condition 7}
 AND C>Ref(C,-1) AND C>O {Condition 8}

increased the pole slope to 28.6% or 2.2% per day.
This I expressed mathematically in terms of the linear

regression slope:

LRS(13) –X1 > 2.2% * C-(X1+13)

Where:
X1=Flag duration in bars or trading days.
LRS-X1=Slope of the 13 bar Linear regression line X1 bars ago.
C-(X1+13) = Closing Price X1+13 bars ago.

Buy condition 3: Slope of flag
The slope of the flag or pennant is limited to no more than 0.2%
and no less than -1.2% per day. According to the statistics in
Figure 1, these values were at the 90% and 10% percentiles,
respectively, and would include 80% of the flags in the sample.
Flags or pennants declining by more than -10% (or -1.2% per
day) were more likely to fail and, hence, excluded.

Buy condition 4: Volume
This condition ensured that volume was declining by more
than -2% per day during the flag or pennant formation and
was implemented in terms of the volume linear regression
line during the formation. The linear regression slope calcu-
lation did not include the signal or current date, as this was
usually outside the formation trendline.

These conditions identified stocks forming a flag or pennant
pattern, but none ensured that it would actually break out
from the formation.

To predict a possible breakout, I compiled a list of 100
flags or similar formations and noted the values and direc-
tion of a number of indicators just before the breakout. I
carried out a frequency distribution analysis and found five
indicators to have predictive power or persistent bullish
readings for more than 90% of the time prior to the breakout
from the formation. These were used in coming up with the
following conditions.

Buy condition 5: Volatility
This condition confirmed that the volatility (as measured by
the standard deviation) was declining during the flag
formation. This was true for more than 90% of the cases
before the breakout.

Buy condition 6: Stochastics and ADX

Both the stochastics oscillator and the average directional
movement index (ADX) had persistently high values before
the breakout, despite the sideways or slightly down price
action. This condition made certain that the 20-day stochastic
was above 55 and the 10-day ADX was above 30.

Buy condition 7: Money flow
Money flow indicators also indicated high readings before
the breakout. More specifically, the finite volume indicator
(FVE) was above 10 for more than 95% of the time and the
volume flow indicator (VFI) was above -3 for more than 90%
of the time. (See sidebar 2, “MetaStock code for the VFI

formula.”) The FVE, as discussed in my September 2003
article, is a short-term money flow indicator. A simplified
interpretation of the FVE is that values above zero indicate a
bullish state and the crossing of the zero line is the buy signal.
The interpretation of the volume flow indicator (VFI) (as
discussed in my June 2004 article) is similar to the FVE; the
only difference is the time horizon under consideration, as
the VFI is a longer-term indicator.

This condition ensured that the 22-day FVE was above 10
and the 130-day VFI was above -3.

Buy condition 8: Price
From the standpoint of logic, the stock should break above
the upper trendline of the formation before initiating a trade.
On testing, however, this condition did not produce the best
results, so I abandoned it. Instead, I used a condition that
required the closing price to be higher than both yesterday’s
close and the opening price.

This confirmed that the stock was starting to wake up from
the brief lethargic state during a flag, but a breakout wasn’t
mandatory.

EXIT CONDITIONS
As with all chart patterns, flags and pennants do not always
perform as expected, so it was necessary to create exit
conditions to preserve profits or minimize losses. Specifying
the optimum exit or stop-loss conditions is sometimes the
most challenging and strenuous task of system design. To
maximize profits and minimize losses, I came up with the
following five exit conditions.

To recreate the VFI in MetaStock, click on the Indicator
Builder (fx), click on “New…”, type “VFI” in the Name box and
paste this code into the formula box.—MK

METASTOCK CODE FOR VFI FORMULA

PERIOD:= Input(“PERIOD FOR VFI “,5,1300,130);
COEF:=.2;
VCOEF:=Input(“MAX VOLUME CUTOFF”,0,50,2.5);
INTER:=Log(Typical())-Log(Ref(Typical(),-1));
VINTER:=Stdev(INTER,30);
CUTOFF:=COEF*VINTER*C;
VAVE:=Ref(Mov(V,PERIOD,S),-1);
VMAX:=VAVE*VCOEF;
VC:=If(V<VMAX,V,VMAX);
MF:=Typical()-Ref(Typical(),-1);
VFI:=Sum(If(MF>CUTOFF, +VC, If(MF <-CUTOFF, -VC,0)),PERIOD)/
VAVE;
Mov(VFI,3,E);

Specifying the optimum exit or
stop-loss conditions is sometimes
the most challenging and strenuous
task of system design.

FIGURE 2: BLUE COAT SYSTEMS (BCSI). Note the high values of stochastic (third window from the top), the VFI (first window from the top in green), and the FVE (second
window from the top) during each trade. Volume also declined during the formation (depicted by the linear regression line in blue). Trade numbers are in green circles.
The first two trades are not shown.

Exit condition 1: Price target
I used the following formula derived in my April 2005 article
to exit the trade:

Formula 1: Target= 1.94*pole0.724

The position was exited on the next bar if the profit
percentage exceeded the value derived by this formula, in
terms of the flagpole. The formula was derived statisti-
cally by means of linear regression on 100 flag and
pennant samples. It was considerably less optimistic than
the traditional measuring formula of equal pole and
breakout percentages.

Exit condition 2: Stop-loss
I used a stop-loss to exit the trade if the stock broke the lower
formation trendline on a closing basis.

Exit condition 3: Inactivity minimum change
This stop was included to stop out trades that did not perform
as expected after a reasonable amount of time. It was trig-
gered if the minimum favorable advance was less than 25%

of the profit target. This stop was set up when the position
was held for more than 14 trading days.

Exit condition 4: Trailing stop-loss
The main reason for including a trailing stop was to preserve
unrealized profits. It was particularly useful when stocks
broke out of the formation but failed to achieve the predicted
profit target and reversed to the downside. This stop was
triggered if the difference between the highest value in the
preceding four-day trailing period and the current bar on a
closing basis was more than 10%.

Exit condition 5: Time exit
If none of these conditions were triggered, the trade was
terminated after 24 trading days. According to the statistics in
Figure 1, this allowed adequate time to include 95% of the
formation breakouts.

If one of these exit conditions was triggered, all positions
were exited at the next day’s close.



  Copyright (c) Technical Analysis Inc.

Stocks & Commodities V. 23:5 (48-57): Detecting Breakouts From Flags & Pennants by Markos Katsanos

DIAGNOSIS
Flag formations and their variations appeared more fre-
quently in volatile technology stocks trading on the Nasdaq.
Nevertheless, for the sake of balance, I chose a sample of four
stocks, two trading on the Nasdaq and two on the New York
Stock Exchange (NYSE).

The system did not compound profits and all trades were
based on a maximum total transaction cost of $10,000. All
orders were executed at the next day’s opening price at the
market, and the commission charged was $10. Interest was
credited at an average rate of 3% when out of the market.

The duration of the test was five years (December 4,
2000, to January 14, 2005) and included a variety of market
conditions, from the devastating 2000–02 bear market to the
2003–04 rally and the most recent sideways market.

TEST RESULTS
Example 1: The first stock I tested was Blue Coat Systems
(BCSI), an Internet software company. I chose this stock
because of the many diverse formation variations and exit
conditions.
     A buy and hold investor would have lost not only his coat

but his shirt and socks as well when the stock dropped 91%
from a high of $207 to $19. Nevertheless, the system was
highly profitable (see Figure 5), yielding an annualized
profit of 45% with no short-selling. The system detected
seven formations including four strings, two pennants, and
one flag. To refresh your memory, the string (as described
in my April 2005 article) is a flaglike formation moving in
a very tight range and making no visible peaks or troughs.
The strings performed better than the pennants, which
verified my statistical results.

Only one trade met or exceeded the profit target predicted
by formula 1, but the elaborate exit conditions managed to
keep the rest of the trades profitable. The first pennant (not
shown in Figure 2) in December 2001 failed to break out,
mainly because of bearish market conditions — and was
exited by the trailing stop-loss (exit condition 4). The second
trade also failed, as the stock was already overextended from
a preceding flag not detected by the system.

The trade was terminated appropriately by the inactivity
exit provision. The third trade (string) did reach the profit
target calculated by formula 1 and was exited on time before
the subsequent correction. The next trade was also triggered

FIGURE 3: APPLIED DIGITAL (ADSX). Only one trade was detected by the system. Note the dramatic drop of the volatility (as depicted by the 10-day standard deviation)
during the formation (first window from the top), the FVE (second window from the top in pink), remained above 30 despite the 12-day selloff. Volume also declined during
the formation (depicted by the linear regression line in blue)

FIGURE 4: GREAT ATLANTIC & PACIFIC TEA CORP (GAP). Only the last three trades are shown. Trade 3 (first trade indicated) missed the predicted target by only
3%. The last trade managed to reach the predicted price objective and was exited in a timely manner, avoiding the abrupt and precipitous reversal that followed.

by a string but was exited prematurely by the stop-loss as it
dipped below the string’s lower trendline. Fortunately, it was
reentered a couple of days later (but at a higher price). This
trade was highly profitable but was a little shy from the profit
target, exited by the inactivity stop.

The sixth trade, initiated by a pennant, was also exited
prematurely by the stop-loss condition as it dipped for only a
day below the pennant’s lower trendline. This was very
unfortunate, as it could have been the test’s best trade had it
not been stopped out. Finally, the last trade, triggered by a
flag, also did not reach the price target and was stopped out
by the trailing stop-loss condition.

Example 2: Applied Digital Solution Corp. (ADSX) formed a
funnel (see Figure 3) after a vehement 118% rise in late October
2004. The funnel is a flaglike variation and looks like a mini–
broadening formation or a pennant turned backward.

In the next 20 days, the stock broke out, exceeding the
projected price target. The Great Atlantic & Pacific Tea Corp.
(GAP) test in Figure 4 detected two strings (not shown) and three
pennants.

Only two trades met or exceeded their price target, while the

other three, although profitable, did not quite reach their respec-
tive price targets and were exited by the inactivity or trailing
stop-loss exit provision. The fourth trade, which used the previ-
ous trade’s breakout for a pole, did not reach the expected target,
as the stock was already exhausted by the preceding successful
pennant breakout.

Example 3: US Gypsum Corp. (not shown) produced rela-
tively frequent flags or pennants, as it tended to move violently
on news of asbestos litigation reform by Congress. This test,
although the least profitable, produced a respectable 69%
profit compared with a 17% loss for the buy and hold investor.
In this case, the trailing stop was detrimental to the overall
performance as it triggered a premature exit from the fourth

Selecting an exit is crucial to a
successful trade. In designing this
system, I tried to remove subjective
human emotions by specifying
automatic exit conditions.

PROFITABILITY REPORT: FLAG & PENNANT SYSTEM

CHART Fig 2 Fig 3 Fig 4 Fig 6
STOCK SYMBOL BCSI ADSX GAP USG Exploration

Total Net Profit $18,725 $11,071 $14,304 $6,888 $26,182
Profit/Loss % 187% 111% 143% 69% 26%
Annual % Gain/Loss 45.5% 22.1% 28.6% 13.8% 735%
Buy and Hold Profit/Loss -$9,094 -$9,339 -$6,934 -$1,705
Buy and Hold % Profit/Loss -91% -93% -69% -17%
Annual Buy and Hold % Profit/Loss -22.1% -18.6% -13.8% -3.4%
Total Number of Trades 7 1 5 5 13
Winning Trades 7 1 5 3 11
Losing Trades 0 0 0 2 2
% Profitable 100% 100% 100% 60% 85%
Avg. Winning Trade $2,368 $9,082 $2,346 $2,403 $2,522
Avg. Losing Trade N/A N/A N/A -$1,235 -$648
Ratio Avg. Win/Avg. Loss N/A N/A N/A 1.95 3.89
Profitability Coefficient 100% 100% 100% 26% 64%
Best Trade $11,598 $9,082 $3,284 $3,681 $6,410
Worst Trade -$2,510 N/A $1,538 -$2,030 -$717
Most Adverse Intraday Excursion -$1,374 -$463 -$651 -$2,160 $0
Max. Equity Drawdown (below initial) $0 $0 $0 $0 $0
Reference Bars Needed 260 260 260 260 260
Start Date/Loaded Data 11/22/99 12/31/98 12/31/98 12/31/98 5/16/02
Start Date/Test 12/4/00 1/13/00 1/13/00 1/13/00 6/2/03
End Date 1/14/05 1/14/05 1/14/05 1/14/05 6/13/03
Test Period (days) 1,502 1,828 1,828 1,828 13
Avg. Trade Length (bars) 12 21 18 12 10
Total Time in the Market (days) 125 32 137 78 13
Total Time % in the Market 8% 2% 7% 4%
Stock Price at Start of Test 207.50 77.80 28.81 42.56
Stock Price at End of Test 19.09 5.24 8.89 35.49

FIGURE 5: PROFITABILITY REPORT. The system produced respectable profits with no short sales, despite abysmal buy and hold
performance. Intraday drawdown was very low mainly because of the elaborate exit conditions. The stop-loss condition did not always
perform as intended, and in the case of USG, it stopped out a trade that could have been the best of the test.

FIGURE 6: RESULTS OF AN EXPLORATION RUN ON 6/1/03. Trades are entered at next day’s opening price and exited manually according to the same exit conditions
used for the test and described in the article. If one of the exit conditions becomes prime, trades are stopped at next day’s close. All trades were limited to 5% of the trading
day’s volume. In case of illiquid stocks (like NOOF), the trade was extended for the next three days until the required amount of shares was accumulated.

EXPLORATION RESULTS
Trade Trade

Entry No. of Total Pole Pole Pole Target Target Exit Exit Duration Duration Profit Reason
Symbol Price Shares Amount Top Bottom % % Price Price Date (Bars) (Days) % Amount for Exit

ATPG 4.51 1,704  $7,685 4.75 3.1 53.2 34.5 6.06 6.75 6/11/03 8 9 49.7%  $11,502 Target
CVM 0.70 10,977  $7,684 0.47 0.21 123.8 63.5 0.74 1.27 6/4/03 2 2 81.4%  $13,941 Target
LNG 2.78 2,763  $7,681 2.89 1.76 64.2 39.5 3.77 5.1 6/11/03 8 9 83.5%  $14,091 Target
EMKR 2.89 2,658  $7,682 2.87 1.8 59.4 37.3 3.57 3.74 6/4/03 2 2 29.4%  $9,941 Target
GLBL 5.70 1,348  $7,684 5.71 4.38 30.4 23.0 6.80 5.27 6/16/03 5 14 -7.5%  $7,104 Stop Loss
GRIC 4.36 1,763  $7,687 4.35 2.35 85.1 48.4 6.09 4.58 6/16/03 11 14 5.0%  $8,075 Trailing Stop
HCR 23.70 324  $7,679 24 19.29 24.4 19.6 28.34 27.05 7/3/03 24 31 14.1%  $ 8,764 Time exit
MSO 11.03 696  $7,677 12 8.11 48.0 32.0 14.40 10 6/4/03 2 2 -9.3%  $6,960 Stop Loss
NOOF 1.26 6,097  $7,682 1.3 0.75 73.3 43.5 1.78 1.69 6/27/03 18 25 34.1%  $10,304 Target
PKD 2.77 2,773  $7,681 2.87 1.96 46.4 31.2 3.66 2.82 6/20/03 15 18 1.8%  $7,820 Inactivity
STEM 1.70 4,519  $7,682 1.82 0.66 175.8 81.9 3.13 1.96 6/17/03 12 15 15.3%  $8,857 Trailing Stop
TKO 2.54 3,026  $7,686 2.68 1.48 81.1 46.8 3.71 3.43 6/9/03 7 7 35.0%  $10,379 Target
USG 11.69 657  $7,680 12 4.98 141.0 69.8 19.76 13.05 6/20/03 15 18 11.6%  $8,574 Inactivity

 $99,870 average 9.9 12.8  $126,312
              Commission      +130  -130

$100,000 26.2% $126,182

Column E
Name: Target %
Formula:
 zz:=Zig(C,17,%);
X:=BarsSince(ZZ<Ref(ZZ,-1)AND Ref(ZZ,-1)>Ref(ZZ,-2));
X1:=LastValue(X)+1;
c2:=Ref(HHV(C,(2+X1)),-1);
c3:=Ref(LLV(C,26),-X1);
POLE:=(C2-C3)/C3*100;POLE
TARGET:=1.94* Power(pole ,.724);TARGET

Column F
Name: Target Price
Formula:
 zz:=Zig(C,17,%);
X:=BarsSince(ZZ<Ref(ZZ,-1)AND Ref(ZZ,-1)>Ref(ZZ,-2));
X1:=LastValue(X)+1;
c2:=Ref(HHV(C,(2+X1)),-1);
c3:=Ref(LLV(C,26),-X1);
POLE:=(C2-C3)/C3*100;POLE
TARGET:=1.94* Power(pole ,.724); (1+TARGET/100)*C

Filter
The filter formula is exactly the same used for the Enter Long Order
of the System Test above

Run the exploration on the desired securities and display the
report. Column A is the prebreakout price, column B is the
highest point on the flag pole, column C is the pole base,
column D is the % pole height, column E is the target % profit,
and column F shows the target price. All prices are on a
closing basis. Columns B–F are not essential for running the
exploration, but they will help to calculate your exit price
target. —MK

trade, which could have been the most
profitable of the test.

The system did not produce enough
signals for frequent trading. This, how-
ever, need not be a problem, since scan-
ning a large number of stocks will pro-
duce more than enough trades.

I tested the system further by scan-
ning (using MetaStock’s Explorer) a
database of 1,250 stocks, which in-
cluded all Standard & Poor’s 500
stocks plus another 750 mainly small-
cap stocks. The number of trades var-
ied according to market conditions.
The average number of trades was
seven to eight under bullish market
conditions, decreasing in sideways
markets and drying out under bearish
conditions. See sidebar 3, “Flag &
pennant exploration.”

I chose to present the results of the
exploration run on June 1, 2003 (Fig-
ure 6), because of the relatively large
number of trades generated by the

Go to the Explorer and choose the New button. Enter the
following formulas for the columns and filter:

Column A
Name: Price
Formula: C

Column B
Name: Pole Top
Formula:
 zz:=Zig(C,17,%);
X:=BarsSince(ZZ<Ref(ZZ,-1)AND Ref(ZZ,-1)>Ref(ZZ,-2));
X1:=LastValue(X)+1;
c2:=Ref(HHV(C,(2+X1)),-1);C2

Column C
Name: Pole Bot
Formula:
 zz:=Zig(C,17,%);
X:=BarsSince(ZZ<Ref(ZZ,-1)AND Ref(ZZ,-1)>Ref(ZZ,-2));
X1:=LastValue(X)+1;
C3:=Ref(LLV(C,26),-X1);C3

Column D
Name: Pole %
Formula:
 zz:=Zig(C,17,%);
X:=BarsSince(ZZ<Ref(ZZ,-1)AND Ref(ZZ,-1)>Ref(ZZ,-2));
X1:=LastValue(X)+1;
c2:=Ref(HHV(C,(2+X1)),-1);
c3:=Ref(LLV(C,26),-X1);
POLE:=(C2-C3)/C3*100;POLE

METASTOCK EXPLORATION FOR FLAG AND PENNANT PATTERN-RECOGNITION SYSTEM

FIGURE 7: EXPLORATION RESULTS. Four of the stocks (ATPG, LNG, CVM, NOOF) exceeded their price target,
EMKR and TKO were right on target, three stocks (HCR, PKD, STEM) missed their target and were stopped on
time by the exit conditions, and USG was late to break out and was stopped by the inactivity exit condition only
two days before the breakout. The GRIC trade was also stopped by the trailing-stop condition five days before the
breakout. MSO and GLBL failed miserably and were appropriately halted by the stop-loss condition, thus averting
heavier losses.

exploration. A hypothetical initial equity of $100,000 was
used to purchase all candidates produced by the exploration
at the next day’s opening price. Trades were exited manually
according to the exit conditions described earlier. The results
were highly profitable, producing a total profit of $26,180
with 11 winning trades and only two losers (Figure 7), but at
no point did the equity dip below the initial amount, because
the losing trades were more than offset by the winners. This
is a major advantage of trading multiple stocks at the same
time. The average trade lasted 13 calendar days, or 10 bars.

CONCLUSION
No system I know of works 100%
of the time. In this case, the usual
reasons for failure were either
adverse market conditions or an
overextended state, due to a pre-
vious successful flag breakout.
    However, considering the tests
were conducted over a long pe-
riod that included all market con-
ditions, the presumption is that
the relative post-performance re-

sults of the method used will continue to produce favorable
results.

Selecting an effective exit is crucial to a successful trade.
In designing this system, I tried to remove subjective human
emotions like fear and greed by specifying automatic exit
conditions. These did not always perform as intended, occa-
sionally stopping out trades that could have had a propitious

outcome, but in most cases they managed to keep losses to a
minimum. Keep in mind that there is always a risk-return
tradeoff, and exit conditions are a compromise between risk
tolerance and profit potential.

Finally, I cannot emphasize enough the advantages of
diversification. By opening multiple trades at the same time
(as was demonstrated by the exploration trades) and assum-
ing a profitable strategy, the occasional losing trades will
always be more than offset by the remaining profitable ones.

Markos Katsanos is a trading system developer and a private
trader.

